
Abstract. In functional programming it is common to write programs as com-
position of other simpler functions. This makes it possible to take advantage
of the well-known benefits of modular programming. However, in many cases,
the resulting programs have efficiency problems caused by the generation of
data structures that are solely used for communication between the intervening
functions in the compositions. Many of those intermediate structures can be
eliminated by an appropriate combination of the codes of the involved functions
using a technique called program fusion. In this work, we propose program
fusion techniques for accumulations, which are recursive functions that use ad-
ditional parameters for keeping intermediate results. Accumulations are known
to be difficult to be fused because of the presence of the additional parameters
and the fact that in many cases results are computed in those parameters. Our
analysis is based on a shortcut fusion approach which turns out to be effective
in the case of accumulations. We present benchmarks that illustrate the impact
of shortcut fusion on accumulations.

1. Introduction

In functional programming it is common to write programs as composition of compara-
tively simple functions. This makes it possible to take advantage of the well-known ben-
efits of modular programming, such as readability and maintainability, but in many cases
the resulting programs have efficiency problems, both in terms of execution time and
memory consumption, due to the construction of intermediate data structures required for
the communication between the functions intervening in the compositions. More efficient
definitions can be obtained by the application of a program transformation technique,
known as fusion or deforestation, aiming at the elimination of these intermediate data
structures.

There exist well-known fusion methods that successfully perform such transformations
for an ample set of functions. However, there is a class of functions that have been
traditionally difficult to manipulate for program fusion. These are the so-called ac-
cumulations, which are functions that use additional parameters where they hold in-
termediate results or even build the final results. The difficulty with these functions
relies in the fact that classical fusion techniques are in general not able to reach the
accumulating parameters when accumulations appear as producer functions (i.e. as
the ones that generate the intermediate structure), and therefore cannot fully elimi-
nate the intermediate data structures in those cases. There are different proposals and
approaches that manage to solve fusion involving accumulations for certain groups
of functions, with different success and simplicity levels. Representative examples



are [Hu et al. 1999, Gibbons 2000, Svenningsson 2002, Nishimura 2002, Pardo 2003,
Voigtländer and Kühnemann 2004, Voigtländer 2004, Katsumata and Nishimura 2006].

In this work we focus on a fusion technique based on an approach known as shortcut fu-
sion [Gill et al. 1993]. We pay special attention to the case when accumulations play the
role of producer function and propose a particular treatment for this case in the style
of shortcut fusion. The solution we present is simple, easy to use and automate. It
is worth mentioning that, like standard shortcut fusion, the technique we introduce is
generic, meaning that it can be defined in an uniform way for an ample class of datatypes.

Along with this proposal, we perform a set of benchmarks allowing running time compar-
isons between the transformed and the original programs, thus helping identify program
groups for which the application of the technique is effective. Throughout we will use
Haskell notation, assuming a cpo semantics (in terms of pointed cpos), but without the
presence of the seq function [Johann and Voigtländer 2004].

The paper is organized as follows. We start in Section 2 with a review of the concept of
shortcut fusion. In Section 3, by means of specific examples, we study the application
of shortcut fusion for programs involving accumulations. The generic formulation of the
concepts and laws developed in Sections 2 and 3 are presented in Section 4. Section 5
presents a benchmark that evaluates the application of shortcut fusion to accumulations.
Finally, Section 6 concludes the paper.

2. Shortcut fusion

Shortcut fusion [Gill et al. 1993, Gill 1996] is a program transformation technique that
allows the elimination of intermediate data structures generated in compositions of the
form c ◦ p by a suitable combination of the definitions of c (the consumer) and p (the
producer). This technique is a consequence of parametricity properties (known as “free
theorems” [Wadler 1989]) associated to polymorphic functions.

For its application, shortcut fusion requires that both the consumer and producer functions
satisfy certain (structural) restrictions. The consumer c is required to process all the
elements of the intermediate data structure in a uniform way. Concretely, this condition is
established by requiring that c is expressible as a fold [Bird 1998], a program scheme that
captures function definitions by structural recursion. Regarding the producer p, it must be
such that the generation of the intermediate data structure is performed using uniquely the
constructors of the data type and not whatever values that are passed to it as arguments.
For example, if p produces a list as result, then it must ensure that the list is generated
only in terms of the list constructors [ ] and (:). To meet this condition the producer is
required to be expressible in terms of a so-called build function [Gill et al. 1993], which
is a function that carries a “template” that exhibits the occurrences of the constructors
of the intermediate datatype. The essential idea of shortcut fusion is then to replace, in
the template, the occurrences of the intermediate datatype’s constructors by appropriate
functions that are specified within the consumer. As a result, one obtains a definition that
computes the same as the original composition c ◦p but without building the intermediate
data structure. Because it fuses compositions of fold with build this transformation is
usually referred to as the fold/build law.

Next, we show the definition of this law for lists and leaf-labelled binary trees.



Lists As mentioned above, the consumer must be a structural recursive definition repre-
sentable as a fold. In the case of lists fold has the following definition:

foldL :: (b, a → b → b) → [a ] → b

foldL (nil , cons) [ ] = nil

foldL (nil , cons) (a : as) = cons a (foldL (nil , cons) as)

This function is equivalent to the well-known foldr function [Bird 1998]. It traverses the
list and replaces [ ] by the constant nil and the occurrences of (:) by function cons . For
example, the function that computes the length of a list:

length [ ] = 0

length (a : as) = 1 + length as

can be written as foldL (0, λa r → 1 + r).

The producer, on the other hand, must be expressible in terms of a build function:

buildL :: (∀ b . (b, a → b → b) → c → b) → c → [a ]

buildL g = g ([ ], (:))

which takes as parameter a function g that abstracts the list constructors from the process
that generates the intermediate list. For example, let us consider the function that given a
list of values of type Maybe a returns a list with the values of type a:

data Maybe a = Nothing | Just a

collect :: [Maybe a ] → [a ]

collect [ ] = [ ]

collect (m : ms) = case m of

Nothing → collect ms

Just a → a : collect ms

It can be written as:

collect = buildL gcoll

where

gcoll (nil , cons) [ ] = nil

gcoll (nil , cons) (m : ms) = case m of

Nothing → gcoll (nil , cons) ms

Just a → cons a (gcoll (nil , cons) ms)

Based on the forms required to the consumer and producer functions it is possible to
define the following shortcut fusion law:
Law 1 (fold/build for lists)

foldL (nil , cons) ◦ buildL g = g (nil , cons)

To see an example of the application of shortcut fusion, consider the definition of function
count = length ◦ collect , which counts the number of elements of type a that occur in



a list of values of type Maybe a. This function produces an intermediate list with the
values of type a. A monolithic definition can be obtained by using Law 1: count =
gcoll (0, λa r → 1 + r). Inlining,

count [ ] = 0

count (m : ms) = case m of

Nothing → count ms

Just a → 1 + count ms

Binary trees Leaf-labelled binary trees can be declared by:

data Btree a = Leaf a | Join (Btree a) (Btree a)

The fold and build functions for this data type are given by:

foldB :: (a → b, b → b → b) → Btree a → b

foldB (leaf , join) (Leaf a) = leaf a

foldB (leaf , join) (Join l r) = join (foldB (leaf , join) l) (foldB (leaf , join) r)

buildB :: (∀ b . (a → b, b → b → b) → c → b) → c → Btree a

buildB g = g (Leaf , Join)

The shortcut fusion law for this type is then the following:

Law 2 (fold/build for binary trees)

foldB (leaf , join) ◦ buildB g = g (leaf , join)

3. Accumulations

Accumulations are recursive functions that use additional parameters for keeping inter-
mediate values or even generating the final result. The use of this kind of functions is
in general motivated by efficiency reasons since in many cases accumulative versions
of functions turn out to be more efficient than nonaccumulative ones. In the context of
program transformation, this fact has been well exploited by the so-called accumulation
technique [Burstall and Darlington 1977, Bird 1998] that transforms recursive definitions
by the introduction of additional arguments where to compute intermediate results.

A prototypical example of the benefits of accumulations is the definition of a linear-time
reverse function. It is well-known that the usual definition of reverse:

reverse :: [a ] → [a ]

reverse [ ] = [ ]

reverse (a : as) = reverse as ++ [a ]

is quadratic in the length of the input list due to the presence of list append (++). In
response to this problem an accumulative linear version can be introduced:1

1or even derived from the nonaccumulative version by the application of the accumulation technique.



reverse as = areverse (as , [ ])

areverse ([ ], xs) = xs

areverse (a : as , xs) = areverse (as , a : xs)

In this section we analyse different situations involving accumulations and the effective-
ness of shortcut fusion in each case. In Section 4 we present the generic formulation of
the different constructions and laws treated here.

3.1. Accumulations as consumers

Accumulations may occur as consumer functions. However, due to the limitation on the
form of the consumer imposed by the fold/build law, shortcut fusion can be applied in
such a case only if the accumulation can be written as a fold . Therefore, as fold does
not handle accumulating parameters, the only alternative is to write the accumulation as a
higher-order fold.

For example, let us consider the accumulative version of the length function:

alength :: [a ] → Int → Int

alength [ ] z = z

alength (a : as) z = alength as (1 + z )

such that length as = alength as 0. Function alength can be written as a fold that returns
a function of type Int → Int :

alength = foldL (id , λa r → r ◦ (1+))

The problem with this definition as a higher-order fold is that it returns a list of suspended
function calls of same dimension as the consumed input list. That is, to compute the length
of the input list, this fold first builds a list of function calls, which is then applied to the
initial value of the accumulator in order to produce the final integer. Such a computation
is therefore less efficient than the one performed by the original definition of alength on
the same input.

Unfortunately, the loss of efficiency caused by the representation of accumulations as
higher-order folds impacts negatively in the performance of the programs that are obtained
when fusing accumulations with producer functions. That is, the resulting fused programs
are expected to be, in general, less efficient than the original ones. This is because, as a
consequence of the application of shortcut fusion, suspended function calls are inserted
in the places corresponding to the constructors within the body of producers. Therefore,
the resulting fused programs produce values of functional type which are isomorphic to
the intermediate data structures eliminated by fusion.

To see an example, recall the definition of function count and suppose we consider the
accumulative definition of length. Then,

count ms = length (collect ms)

= alength (collect ms) 0

= foldL (id , λa r → r ◦ (1+)) (collect ms) 0



= (foldL (id , λa r → r ◦ (1+)) ◦ buildL gcoll) ms 0

= gcoll (id , λa r → r ◦ (1+)) ms 0

Let lc = gcoll (id , λa r → r ◦ (1+)). Inlining,

lc [ ] = id

lc (m : ms) = case m of

Nothing → lc ms

Just a → lc ms ◦ (1+)

In the benchmarks presented in Section 5 we compare execution times between original
and fused programs that involve accumulations. In those benchmarks it can be observed
that in various cases (but not in all ones) less efficient programs are obtained from apply-
ing fusion in a fold/build style to programs that have accumulations as consumers.

In the context of shortcut fusion, there is an alternative treatment of accumulations as
consumers using a dual law called destroy/unfold (see [Svenningsson 2002] for details).

3.2. Accumulations as producers

The effectiveness of fusion in situations where accumulations appear as producer func-
tions depends on the kind of accumulation. In fact, two cases can be identified: (i) when
the accumulating parameters are used to hold auxiliary values that are eventually used in
the final result; and (ii) when the result of the function, or part of it, is constructed in the
accumulating parameters and therefore they are of the same type as the result. Of course,
there may be mixed situations of functions that have some accumulating parameters of
kind (i) and others of kind (ii), but these will simply require a mixed treatment.

In case (i), the fact that the accumulating parameters and the result of the function are of
a different type makes it possible to apply standard shortcut fusion without even being
necessary to identify that the producer is an accumulation.

Let us see an example. Consider the function that computes the height of a binary tree in
two steps: first, it replaces the value of each leaf by its depth, and then it computes the
maximum of the tree.

height :: Btree a → Int

height t = aheight (t , 0)

aheight :: (Btree a, Int) → Int

aheight = maxBtree ◦ depths

maxBtree :: Ord a ⇒ Btree a → a

maxBtree (Leaf a) = a

maxBtree (Join l r) = max (maxBtree l) (maxBtree r)

depths :: (Btree a, Int) → Btree Int

depths (Leaf a, n) = Leaf n

depths (Join l r , n) = Join (depths (l , n + 1)) (depths (r , n + 1))



Since maxBtree can be written as a fold and depths in terms of build ,

maxBtree = foldB (id ,max )

depths = buildB gdep

where

gdep (leaf , join) (Leaf a, n) = leaf n

gdep (leaf , join) (Join l r , n)

= join (gdep (leaf , join) (l , n + 1)) (gdep (leaf , join) (r , n + 1))

by Law 2 we can obtain the following monolithic definition of aheight :

aheight (Leaf a, n) = n

aheight (Join l r , n) = max (aheight (l , n + 1)) (aheight (r , n + 1))

In this case the fact that depths is an accumulation has no impact in the way fusion is
applied. This is because its accumulating parameter holds simply a integer, used to fill
the leaves of the new tree. Therefore, there is nothing that fusion could change on that
parameter; in fact, in the fused program it is used in the same way as it was in depths .

Now, let us analyse case (ii). This is the case where fusion methods have traditionally
failed because of the difficulties to reach the accumulating parameters. The fact that
accumulations compute their results in their accumulating parameters makes it necessary
that fusion takes place also in those parameters. Accumulation in this category are, for
example, function areverse, shown previously, and aflatten, which lists the leaves of a
binary tree in left-to-right order.

aflatten :: (Btree a, [a ]) → [a ]

aflatten (Leaf a, xs) = a : xs

aflatten (Join l r , xs) = aflatten (l , aflatten (r , xs))

To see a simple example of the difficulty in reaching the accumulating parameters, let us
consider the following definition of a function that given a number represented as a list of
digits dn · · · d1d0 returns the corresponding integer.

number :: [Int ] → Int

number ds = anumber (ds , [ ])

anumber = horner ◦ areverse

horner [ ] = 0

horner (d : ds) = d + 10 ∗ horner ds

First, let us try to derive a recursive definition for anumber by case analysis:

anumber ([ ], xs) = horner (areverse ([ ], xs))

= horner xs

anumber (d : ds , xs) = horner (areverse (d : ds , xs))



= horner (areverse (ds , d : xs))

= anumber (ds , d : xs)

The definition obtained by this method is not satisfactory as it continues constructing an
intermediate list (now internal to the function), holding the reverse of the list of digits,
which still needs to be evaluated by the horner function once the base case of the empty
list of digits is reached.

The problem with a direct approach like this is that it does not notice that the accumulat-
ing parameter should have a different type in the fused program. In fact, in this particular
example, the accumulator should change from type [Int] to type Int in the fused program
as it needs to hold the (partial) value of the integer being computed. That is, after apply-
ing fusion one should obtain the following definition, which is the natural tail recursive
definition to this problem,

anumber ′ :: ([Int ], Int) → Int

anumber ′ ([ ], z ) = z

anumber ′ (d : ds , z ) = anumber ′ (ds , d + 10 ∗ z )

together with the equation:

anumber (ds , xs) = anumber ′ (ds , horner xs)

which states that the initial value of the “old” accumulator (the one of the producer) needs
to converted to the format of the “new” accumulator (the one of the fused program). This
conversion is performed by the consumer, the horner function.

In fact, the same equation holds in general for any producer given by an accumulation
of this category. That is, given a consumer c and an accumulative producer p, its fusion
fus satisfies that c ◦ p = fus ◦ (id × c), where (f × g) (x , y) = (f x , g y). This is the
essential equation one has to have in mind when performing fusion involving this kind of
accumulations.

The transformation shown above can be performed using shortcut fusion. The build func-
tion must now abstract the constructors that occur in the accumulating parameters in order
to make them reachable for fusion. For the example, function areverse can be written in
terms of build as follows:

areverse = buildL grev

where

grev (nil , cons) ([ ], xs) = xs

grev (nil , cons) (a : as , xs) = grev (nil , cons) (as , cons a xs)

However, this definition is not complete. In fact, to make the desired transformation
possible the build function needs to include as part of its definition the format conversion
of the initial value of the accumulator, which is performed by the consumer function. So
we need to insert a call to the consumer (a fold ) as part of the producer. The correct
definition is thus as follows:



areverse = buildL grev

where

grev (nil , cons) (as , xs) = grev ′ (nil , cons) (as , foldL (nil , cons) xs)

grev ′ (nil , cons) ([ ], z ) = z

grev ′ (nil , cons) (a : as , z ) = grev ′ (nil , cons) (as , cons a z )

If we apply Law 1 we obtain the definitions of anumber and anumber ′ showed above.

An unsatisfactory aspect of the previous transformation in terms of shortcut fusion is the
fact that the conversion that needs to be done on the accumulator of the producer is in
all cases the same. That, it is always necessary to insert an application of the consumer
function to the initial value of the producer´s accumulator as part of the definition of the
build function. This means that to write the definition of the build function we have
to inspect the definition of the producer in order to infer in which places to insert the
applications of the consumer function. However, it is worth noting that it is possible to
obtain automatically such applications of the consumer if we slightly modify the type of
the template g carried by the build function. Concretely, the modification we propose is
to make explicit in g the fact that the accumulating parameters are of the same type as the
result of g . For instance, in the case of function areverse, instead of having grev of type,

grev :: (∀ b . (b, a → b → b) → ([a ], [a ]) → b)

our proposal is to manipulate a grev function of type

grev :: (∀ b . (b, a → b → b) → ([a ], b) → b)

Having this modification, the desired applications of the consumer to the accumulators
will appear automatically with no effort at all as a consequence of the free theorems
associated with the new type of g . This implies of course the introduction of a modified
version of build that we call builda. Next we show the definition of builda for lists and
leaf-labelled binary trees.

In the case of lists the new build function has the following definition:

buildaL :: (∀ b . (b, a → b → b) → (c, b) → b) → (c, [a ]) → [a ]

buildaL g = g ([ ], (:))

and an associated shortcut fusion law.

Law 3 (fold/builda for lists)

foldL (nil , cons) ◦ buildaL g = g (nil , cons) ◦ (id × foldL (nil , cons))

Writing areverse in terms of builda,

areverse = builda grev

where

grev (nil , cons) ([ ], z ) = z

grev (nil , cons) (a : as , z ) = grev (nil , cons) (as , cons a z )



the same definitions of anumber and anumber ′ seen before are obtained now by the
application of Law 3.

For binary trees, builda has the following definition:

buildaB :: (∀ b . (a → b, b → b → b) → (c, b) → b) → (c,Btree a) → Btree a

buildaB g = g (Leaf , Join)

with a corresponding shortcut fusion law.

Law 4 (fold/builda for binary trees)

foldB (leaf , join) ◦ buildaB g = g (leaf , join) ◦ (id × foldB (leaf , join))

A fold/builda law for similar binary trees, but which do not carry information in the leaves,
i.e. a kind of shape trees, can be found in [Katsumata and Nishimura 2006].

To see an example, consider the following definitions:

printAsc :: (Btree a,Btree a) → Btree a

printAsc = b2s ◦ asc

b2s :: Btree a → String

b2s (Leaf a) = "Leaf "++ show a

b2s (Join l r) = "Join ("++ b2s l ++ ") ("++ b2s r ++ ")"

asc :: (Btree a,Btree a) → Btree a

asc (Leaf a, t) = Join t (Leaf a)

asc (Join l r , t) = asc (l , asc (r , t))

Since b2s can be written as a fold and asc in terms of builda:

b2s = foldB (fleaf , fjoin)

where

fleaf a = "Leaf "++ show a

fjoin x y = "Join ("++ x ++ ") ("++ y ++ ")"

asc = buildaB gasc

where

gasc (leaf , join) (Leaf a, t) = join t (leaf a)

gasc (leaf , join) (Join l r , t) = gasc (leaf , join) (l , gasc (leaf , join) (r , t))

we can apply Law 4 to fuse the parts, obtaining the following definition for printAsc,

printAsc (t , u) = printAsc ′ (t , b2s u)

printAsc ′ :: (Btree a, String) → String

printAsc ′ (Leaf a, s) = "Join ("++ s ++ ") (Leaf"++ show a ++ ")"
printAsc ′ (Join l r , s) = printAsc ′ (l , printAsc ′ (r , s))



4. Formalization
In this section, we show that the instances of fold , build , and shortcut fusion presented in
the previous sections correspond to generic definitions valid for a wide class of datatypes.

4.1. Data types

The structure of datatypes can be captured using the concept of a functor. A functor
consists of a type constructor F and a function mapF :: (a → b) → (F a → F b), which
preserves identities and compositions: mapF id = id and mapF (f ◦ g) = mapF f ◦
mapF g . A standard example of a functor is that formed by the list type constructor and
the well-known map function.

Semantically, recursive datatypes are understood as least fixed points of functors. That is,
given a datatype declaration it is possible to derive a functor F such that the datatype is
the least solution to the equation τ ∼= Fτ . We write µF to denote the type corresponding
to the least solution. The isomorphism between µF and F µF is provided by two strict
functions inF :: F µF → µF and outF :: µF → F µF , inverses of each other. Function
inF packs the constructors of the datatype while outF the destructors (for more details
see e.g [Abramsky and Jung 1994, Gibbons 2002]).

In the case of lists, the structure is captured by a bifunctor L (a functor on two variables)
because of the presence of the type paremeter. That is, µ(La) ∼= [a ].

data L a b = FNil | FCons a b

mapL :: (a → c) → (b → d) → L a b → L c d

mapL f g FNil = FNil

mapL f g (FCons a b) = FCons (f a) (g b)

4.2. Fold

Let F be a functor that captures the structure of a datatype. Given a function ϕ::F a → a,
fold [Gibbons 2002] is defined as the least function foldF ϕ :: µF → a such that:

foldF ϕ ◦ inF = ϕ ◦ F (foldF ϕ)

A function ϕ :: F a → a is called an F -algebra. For example, an algebra corresponding
to the functor L a is a function ϕ :: L a b → b of the form:

ϕ FNil = nil and ϕ (FCons a b) = cons a b

with nil :: b and cons :: a → b → b. In the specific instance of fold for the list datatype
[a] we wrote an algebra ϕ simply as a pair (nil , cons). The same can be applied to other
datatypes.

4.3. Shortcut fusion

Given a functor F , we can define a corresponding build operator that captures producer
functions that generate structures of type µF .

buildF :: (∀ a . (F a → a) → c → a) → c → µF

buildF g = g inF



Notice that the abstraction of the datatype’s constructors is given in terms of an F-algebra.
Together with fold , build enjoys the following fusion law [Takano and Meijer 1995],
which is an instance of a free theorem [Wadler 1989].

Law 5 (fold/build) For strict ϕ,2

foldF ϕ ◦ buildF g = g ϕ

4.4. Shortcut fusion for accumulations

In the same way as build , a generic definition of the builda function that captures ac-
cumulative producers can be formulated. The only difference with build is the explicit
occurrence of the accumulator (of the same type as the result).

buildaF :: (∀ a . (F a → a) → (c, a) → a) → (c, µF ) → µF

buildaF g = g inF

The shortcut fusion law in this case is the following.

Law 6 (fold/builda) For strict ϕ,

foldF ϕ ◦ buildaF g = g ϕ ◦ (id × foldF ϕ)

Proof The free theorem associated with g’s type states that, for all types a and b, algebras
ψ :: F a → a and ϕ :: F b → b, and strict function f :: a → b, the following holds
f ◦ ψ = ϕ ◦ mapF f ⇒ f ◦ g ψ = g ϕ ◦ (id × f ). By considering f = foldF ϕ and
ψ = inF , we get foldF ϕ ◦ g inF = g ϕ ◦ (id × foldF ϕ), because the premise of the
implication holds by definition of fold. Finally, by applying the definition of buildaF we
obtain the law. The strictness on ϕ is necessary for instantiation: if the algebra ϕ is strict,
then so is foldF ϕ, and we can instantiate f with foldF ϕ. 2

5. Benchmarks
With the aim at illustrating the impact of shortcut fusion on accumulations we present
a benchmark comparing time performance of a set of example programs. The programs
considered in the benchmark are simple but representative of different cases involving
accumulations. All of them have an accumulation as producer function, and in some
cases accumulations also appear as consumer functions.

Figure 1 presents the speedup percentages of comparing the running times of the original
programs against those obtained by the application of shortcut fusion. The results were
obtained by compiling the programs with the Glasgow Haskell Compiler (GHC) 6.8.2. In
order to measure the effects of shortcut fusion in isolation, we compiled both the original
and the transformed programs without activating the native optimisations of GHC. The
transformed programs correspond to the recursive definitions that are obtained by inlining
the expressions that result from the application of shortcut fusion. Those definitions were
generated by hand.

The name of the programs considered in Figure 1 are formed by the concatenation of the
names of the consumer and the producer functions, like e.g. “LengthFlatten”. Consumer

2The strictness condition on ϕ was not mentioned in the concrete instances of the law shown in Section 2
because the algebras considered in those instances are all strict.

names with suffix ‘Ac’ correspond to accumulations. That is the case of “LengthAc”, for
example. The names of most programs are self-explanatory, except for those with suffix
“Rp”; this suffix corresponds to a function on shape trees (binary trees without informa-
tion in the leaves) that replaces the rightmost leaf by a tree. The leftmost ten programs
have accumulations as consumers. With the exception of program “HeightAcAsc”, the
application of shortcut fusion to those programs results in slower ones as expected. On
the other hand, for the programs with nonaccumulative functions as consumers (the right-
most eight ones), out of “MapFlatten”, the application of shortcut fusion results in more
efficient programs.

A prototype implementation of shortcut fusion has been obtained using the rewrite rules
mechanism (the RULES pragma) of GHC, which permits the specification of equational
transformations within programs that are applied by the compiler during optimisation. To
apply the rewrite rules it is necessary to write the consumer functions as a fold, and the
producer functions as a build. So far this has been done by hand. We believe that the
development of a procedure for the automatic derivation of fold and build representations
for accumulations in the line of warm fusion [Launchbury and Sheard 1995] is possible.
We performed some experiments using rewrite rules, but we do not present their results
because the interplay between shortcut fusion and other optimisations in the context of
accumulations deserves further analysis. In fact, some of the results have certain similarity
with those presented in Figure 1, but there are others that are dramatically different.

6. Conclusions

This paper presented an study of the application of shortcut fusion to programs involving
accumulations. In particular, we introduced a modified version of the fold/build law tai-
lored to the cases when producer functions are given by accumulations that compute their
results (or part of them) in accumulating parameters. The new law required the definition
of a slightly modified build function, called builda, that explicitly exhibits in its type the
occurrences of those accumulating parameters. Benchmarks showed the effectiveness of
the new law for a variety of example programs.

